
1   Change 1 - function f_TC_6_3_4_UTRAN
	Function name
	function f_TC_6_3_4_UTRAN()

	Reason for change
	Test case hung after step 0S as a result of the TTCN sends Authentication parameters twice. Once before and once after the transition to UTRAN PCH state( ie authenticaiton parameters already sent in f_UTRAN_UE_FACH_PS_Rb_Est-> f_UTRAN_UE_PsRB_Est_InitialRegistration-> f_UTRAN_UE_PsRB_Est)


	Summary of change
	Remove redundant send authentication parameters function.


	TTCN module
	LTE_IRAT\source\LTE_IRAT\6_3\CSG_TestcasesUG.ttcn


Before change

  function f_TC_6_3_4_UTRAN() runs on UTRAN_IRAT_PTC

  {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := 

    :

    :
    //@siclog "Step 0R" siclog@

    // @sic R5-155979 sic@

    //Steps 1 to 8 of the Routing Area Update procedure described in TS 36.508 subclause 6.4.2.8 are performed on Cell 5. NOTE: The UE performs RAU

    f_UTRAN_RRC_ConnEst(utran_Cell5);

    f_UTRAN_GMM_RAU_MappedContext(utran_Cell5);

    f_UTRAN_RRC_ConnRel (utran_Cell5, cell_Fach_Dcch); //@sic R5s160066 sic@

    //@siclog "Step 0S" siclog@

    // @sic R5-160926 sic@

    //The Procedure for bringing the UE to cell_PCH state defined in TS34.108 clause 7.4.2.7 takes place and the UE is moved to CELL-PCH state

    f_UTRAN_UE_FACH_PS_Rb_Est(utran_Cell5, waitForIMS); //@sic R5s160066 sic@

    f_UTRAN_TransitToCellPCH (utran_Cell5, 8); //@sic R5s160066 sic@

    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_UTRAN_TestBody_Set (false);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //Release the UTRA cells

    f_UTRAN_ReleaseCell (utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5));

  }//end of 6.3.4
After change
  function f_TC_6_3_4_UTRAN() runs on UTRAN_IRAT_PTC

  {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := 

    :

    :
    //@siclog "Step 0R" siclog@

    // @sic R5-155979 sic@

    //Steps 1 to 8 of the Routing Area Update procedure described in TS 36.508 subclause 6.4.2.8 are performed on Cell 5. NOTE: The UE performs RAU

    f_UTRAN_RRC_ConnEst(utran_Cell5);

    f_UTRAN_GMM_RAU_MappedContext(utran_Cell5);

    f_UTRAN_RRC_ConnRel (utran_Cell5, cell_Fach_Dcch); //@sic R5s160066 sic@

    //@siclog "Step 0S" siclog@

    // @sic R5-160926 sic@

    //The Procedure for bringing the UE to cell_PCH state defined in TS34.108 clause 7.4.2.7 takes place and the UE is moved to CELL-PCH state

    f_UTRAN_UE_FACH_PS_Rb_Est(utran_Cell5, waitForIMS); //@sic R5s160066 sic@

    f_UTRAN_TransitToCellPCH (utran_Cell5, 8); //@sic R5s160066 sic@

    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC

    // Anritsu_CSG_634 - Authentication have been send by the f_UTRAN_UE_FACH_PS_Rb_Est already
    // remove - f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_UTRAN_TestBody_Set (false);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //Release the UTRA cells

    f_UTRAN_ReleaseCell (utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5));

  }//end of 6.3.4
2 Change 2 - function f_TC_6_3_4_EUTRA
	Function name
	function f_TC_6_3_4_EUTRA()

	Reason for change
	Test case failed at step 4/TAU procedure as a result of incorrect epsBearerContextStatus template used by the TTCN. (See Note 1)

Note 1: 2 radio bearers are established ( bearer ID 5 and ID 6) prior to test step 4.But  the function call    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT uses an incorrect epsBearerContextStatus template containing a single radio bear ID.



	Summary of change
	Add new function call f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT_2Bearers with correct template setting.
See Change 2.1

	TTCN module
	LTE_IRAT\source\LTE_IRAT\6_3\CSG_TestcasesUG.ttcn


Before change

  function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList_T0;

    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;

    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;

    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;// @sic R5-155979 sic@

    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;// @sic R5-155979 sic@
    :

    :
    //@siclog "Step 3" siclog@

    //The SS changes Cell 1, Cell 2 levels according to the row "T2" in table 6.3.4.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);  //@sic R5-155979 sic@

    t_Wait.start;

    //@siclog "Step 4-10" siclog@

    //Check: Does the test result of steps 1 to 7 of the generic test procedure in TS 36.508 table

    //6.4.2.7A-1 indicate that the UE is camped on E-UTRAN Cell 2 within 6 mins?

    // PDP Context established in UTRA

    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT(eutra_Cell2, v_CSInOtherRAT, true, false, v_IRAT_Coordination_MSG.OctetData, NORMAL);
    t_Wait.stop; // If this times out it will be caught in the default behaviour

    //Tell UTRA PTC that test case is over

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    f_EUTRA_TestBody_Set (false);

    //@sic R5s150804 sic@

    f_EUTRA_SetCellPower(eutra_Cell1,tsc_NonSuitableCellRS_EPRE );

    f_EUTRA_Postamble_EmptyCSG_AllowedList (eutra_Cell2, eutra_Cell2, {v_CSG_Identity_Cell2}, E1_IDLE); //@sic R5s150983 sic@

    //@sic R5s160066 sic@

  } // end of 6.3.4  function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC
After change
  function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList_T0;

    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;

    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;

    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;// @sic R5-155979 sic@

    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;// @sic R5-155979 sic@
    :

    :
    //@siclog "Step 3" siclog@

    //The SS changes Cell 1, Cell 2 levels according to the row "T2" in table 6.3.4.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);  //@sic R5-155979 sic@

    t_Wait.start;

    //@siclog "Step 4-10" siclog@

    //Check: Does the test result of steps 1 to 7 of the generic test procedure in TS 36.508 table

    //6.4.2.7A-1 indicate that the UE is camped on E-UTRAN Cell 2 within 6 mins?

    // PDP Context established in UTRA

    // Anritsu_CSG_634

    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT_2Bearers (eutra_Cell2, v_CSInOtherRAT, true, true, v_IRAT_Coordination_MSG.OctetData, NORMAL);
    t_Wait.stop; // If this times out it will be caught in the default behaviour

    //Tell UTRA PTC that test case is over

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    f_EUTRA_TestBody_Set (false);

    //@sic R5s150804 sic@

    f_EUTRA_SetCellPower(eutra_Cell1,tsc_NonSuitableCellRS_EPRE );

    f_EUTRA_Postamble_EmptyCSG_AllowedList (eutra_Cell2, eutra_Cell2, {v_CSG_Identity_Cell2}, E1_IDLE); //@sic R5s150983 sic@

    //@sic R5s160066 sic@

  } // end of 6.3.4  function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC
Add function Call - f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT_2Bearers
	Function name
	function f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT_2Bearers

	Summary of change
	Requires new function call f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers. See 2.2.

	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn


Before change

  function f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT_2Bearers(EUTRA_CellId_Type p_CellId,

                                                                        boolean p_CSInOtherRAT,

                                                                        boolean p_PDPContext,

                                                                        boolean p_PreambleOnLTE,

                                                                        template (omit) octetstring p_EUTRA_Cap,   // @sic R5-123734 sic@

                                                                        EUTRA_ATTACH_Type p_ForcedAttach,

                                                                        RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def) runs on EUTRA_PTC

  { /* Checks whether the UE is camping on a new E-UTRAN cell with different TAI of a test case or not.

       Procedure is defined in 36.508 cl. 6.4.2.7A */

    // @sic CR R5-102241 sic@

    // Anritsu_CSG_634

    f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers(p_CellId, p_RRC_TI, p_CSInOtherRAT, p_PDPContext, p_PreambleOnLTE, p_EUTRA_Cap, -, p_ForcedAttach);    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "508Check CampOnNewEutraCellUponMobilityfromAnotherRAT");

  }  
Add  function Call - function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers
	Function name
	function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers

	Summary of change
	Requires new function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq_2Bearers. See 2.3.

	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn


Before change
  function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers(EUTRA_CellId_Type p_CellId,

                                                    RRC_TransactionIdentifier p_RRC_TI,

                                                    boolean p_CSInOtherRAT,

                                                    boolean p_PDPContext,

                                                    boolean p_PreambleOnLTE,

                                                    template (omit) octetstring p_EUTRA_Cap,   // @sic R5-123734 sic@

                                                    RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease, // @sic R5s150830 sic@

                                                    EUTRA_ATTACH_Type p_ForcedAttach := NORMAL) runs on EUTRA_PTC

  { /* @sic R5s120027 Additional Changes sic@ */

    var EstablishmentCause v_EstablishmentCause := mo_Signalling;

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(p_CellId, v_EstablishmentCause);    // @sic R5-120721 change 3 sic@

    f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq_2Bearers(p_CellId,

                                                                p_RRC_TI,

                                                                p_CSInOtherRAT,

                                                                p_PDPContext,

                                                                p_PreambleOnLTE,

                                                                p_EUTRA_Cap,   // @sic R5-123734 sic@

                                                                p_ReleaseRequired,

                                                                p_ForcedAttach);

  }   

Add  function Call - function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_2Bearers
	Function name
	f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq_2Bearers

	Summary of change
	Add Bear ID 6 to epsBearerContextStatus template

	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttc


Before change
function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq_2Bearers

                                                                        (EUTRA_CellId_Type p_CellId,

                                                                         RRC_TransactionIdentifier p_RRC_TI,

                                                                         boolean p_CSInOtherRAT,

                                                                         boolean p_PDPContext,

                                                                         boolean p_PreambleOnLTE,

                                                                         template (omit) octetstring p_EUTRA_Cap,

                                                                         RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease,

                                                                         EUTRA_ATTACH_Type p_ForcedAttach := NORMAL) runs on EUTRA_PTC return template (omit) NasCount_Type

    { /* @sic R5s120027 Additional Changes sic@ */

      var NAS_MSG_Indication_Type v_NasInd;

      var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

      var template (omit) NasCount_Type v_NasCountUL := omit; // @sic R5s120944 sic@

      var NasCountInfo_Type v_NasCountInfo;

      var SRB_COMMON_IND v_ReceivedAsp;

      var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);

      var B3_Type v_EpsUpdate_Result := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@

      var template (present) B3_Type v_EpsUpdate_TypeValue := v_EpsUpdate_Result; // To be used in Request message

      var template (present) EPS_BearerContextStatus v_EPS_ContextStatus_Rx := cr_508_EPS_BearerContextStatusTAUReq;

      var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus_Tx;

      var template (present) NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5s110176 sic@

      var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (p_ForcedAttach); // @sic R5s110176 sic@

      var template (present) NAS_MSG_Indication_Type v_NAS_Expected, v_TAU_Expected, v_TAU_NoProtection, v_TAU_StartInOtherRAT, v_TAU_ForcedMapped;

      var boolean v_NoAuthSincePowerOn;

      var template (omit) AccessStratumRelease v_Rel := omit;

      var UE_EUTRA_Capability v_EUTRA_Cap;

      var boolean v_Rel8_9 := false;

      var  template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (p_ForcedAttach);

      var B3_Type v_PDNType := f_GetPdnType();

      var template CiphKeySeqNum v_CiphKeySeqNum := cr_CiphKeySeqNum('8'H, ?) ifpresent;

      timer t_WaitforAttach := f_EUTRA_SetTimerToleranceMax( p_CellId, nonProtocolTimer, tsc_WaitforAttach);  // @sic R5-134358 sic@

      if (p_PreambleOnLTE) { // @sic R5-123734 sic@

        v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease ();

      } else {

        if (ispresent(p_EUTRA_Cap)) {

          v_EUTRA_Cap := f_EUTRA_DecodeEutraCapMsg(valueof(p_EUTRA_Cap)); // This needs to be a template as it can be omitted

          v_Rel := v_EUTRA_Cap.accessStratumRelease;

        }

      }

      if (not p_PDPContext and not p_PreambleOnLTE) { // no bearer has previously been established

        v_EPS_ContextStatus_Rx.ebi5_7 := '000'B; // so set all bearer indications to 0

      }

      else {

        if (p_PDPContext and p_PreambleOnLTE) {

          v_EPS_ContextStatus_Rx.ebi5_7 := '011'B;
        }

      }

      if (not match(tsc_EpsUpdate_TaUpdate, v_EpsUpdate_TypeValue)) { // TA Only;  @sic R5-110746 sic@

        if (p_CSInOtherRAT) {

          v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_IMSIAttach;

        } else {

          v_EpsUpdate_TypeValue := (tsc_EpsUpdate_Combined_TaLaUpdate, tsc_EpsUpdate_Combined_IMSIAttach); // @sic R5s120943, R5-140325 sic@

        }

      }

      // @sic R5s110168 change 8.5 sic@

      v_TAU_NoProtection := cr_NAS_Indication(tsc_SHT_NoSecurityProtection,

                                              cr_TAU_RequestIdleMobilityDiffRAT(v_EpsUpdate_TypeValue,

                                                                                '111'B,

                                                                                v_EPS_ContextStatus_Rx,

                                                                                v_AdditionalUpdateType_Expected,

                                                                                 ?,

                                                                                 v_CiphKeySeqNum));

      v_TAU_StartInOtherRAT := cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                 cr_TAU_RequestIdleMobilityDiffRAT(v_EpsUpdate_TypeValue,

                                                                                   ?,

                                                                                   v_EPS_ContextStatus_Rx,

                                                                                   v_AdditionalUpdateType_Expected,

                                                                                   ?,

                                                                                   v_CiphKeySeqNum));

      v_TAU_ForcedMapped := cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                              cr_TAU_RequestIdleMobilityDiffRAT(v_EpsUpdate_TypeValue,

                                                                                f_EUTRA_SecurityKSIsgsn_Get (),

                                                                                v_EPS_ContextStatus_Rx,

                                                                                v_AdditionalUpdateType_Expected,

                                                                                tsc_NasKsi_MappedSecurityContext,

                                                                                v_CiphKeySeqNum));

      /* Step 1 - 3 */

      /*  ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN )

          Check that TRACKING AREA UPDATE REQUEST message is present and conforms

          with default message ( 36.508 Table 4.7.2-27: TRACKING AREA UPDATE REQUEST ) */

      if (v_SecurityParams.KSIasme == '111'B) { // Did not do authentication on EUTRA cell since power on

        v_NoAuthSincePowerOn := true;

        if (v_SecurityParams.KSIsgsn == '111'B) {// No current EPS security context stored

          //@sic R5-110746 Don't care if UE is sending mapped or native sic@

          v_TAU_Expected := (v_TAU_NoProtection, v_TAU_StartInOtherRAT);

        }

        else {  /* This should be when: - UE started on 3G/2G,

                 *                      - came to LTE and was forced to use the mapped context in an SMC on LTE

                 *                      - went back to 3G/2G

                 *                      - is now back on LTE */

          v_TAU_Expected := v_TAU_ForcedMapped;

        }

      }

      else { // UE and SS still have current EPS NAS Security Context because previously registered on LTE so should have a valid native context

        v_NoAuthSincePowerOn := false;

        v_TAU_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                            cr_TAU_RequestIdleMobilityDiffRAT(v_EpsUpdate_TypeValue,

                                                                              f_EUTRA_SecurityKSIasme_Get(),

                                                                              v_EPS_ContextStatus_Rx,

                                                                              v_AdditionalUpdateType_Expected,

                                                                              tsc_NasKsi_NativeSecurityContext)); // @sic R5-110746 sic@

      }

      if (p_PDPContext or p_PreambleOnLTE) { // bearer has previously been established

        v_NAS_Expected := v_TAU_Expected;

      } else  { // no bearer established, so behaviour is optional

        if (ispresent(v_Rel)) { // @sic R5-123734 sic@

          if (valueof(v_Rel) == rel8 or valueof(v_Rel) == rel9) {

            v_NAS_Expected := (v_TAU_Expected, cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                                                 cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                 cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType))); // @sic R5s120659 sic@

          } else { // rel10 and above

            v_NAS_Expected := cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                                cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)); // @sic R5s120659 sic@

          }

        } else { // started on GERAN so expect anything for now

          v_NAS_Expected := (v_TAU_Expected, cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                                               cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                               cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType))); // @sic R5s120659 sic@

        }

      }

      //Send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

      f_EUTRA_RRC_ConnectionSetup_Def (p_CellId, -, p_RRC_TI);

      //Receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):

      v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(p_CellId,                                       // @sic R5s120121 MCC160 implementation sic@

                                                         cr_508_RRCConnectionSetupComplete(p_RRC_TI, ?, -, 1),

                                                         v_NAS_Expected);  // @sic R5-123734 sic@

      if (ischosen (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST)) {   // @sic R5-123734 sic@

        v_SecurityParams.NonceUE := oct2bit(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.nonce.nonceValue);

      }

      if (p_PDPContext or p_PreambleOnLTE) { // bearer already established, so go ahead and complete the TAU; @sic R5-110746, R5s110087 sic@

        if (v_NoAuthSincePowerOn) {

          v_SecurityParams := f_EUTRA_MappedContext_InitNAS(v_SecurityParams, v_SecurityParams.AuthParams);

          v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST); // @sic R5s110210 sic@

          /* Step 4, 5: NAS Security Mode Command / NAS Security Mode Complete */

          v_NasCountUL := f_EUTRA_NAS_ActivateSecurity_MappedContext(p_CellId, v_SecurityParams);

        } else { // @sic R5s130238 sic@

          v_NasCountInfo := f_EUTRA_SS_NAS_CountGet();

          v_NasCountUL := v_NasCountInfo.UL;

        }

        v_EPS_ContextStatus_Tx := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

        v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST);

        f_EUTRA_Security_Set(v_SecurityParams);

        /* Step 6: Transmit TRACKING AREA UPDATE ACCEPT message */

        /* Step 7: Receive TRACKING AREA UPDATE COMPLETE */

        f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,

                                                   v_GutiParams,

                                                   v_EpsUpdate_Result,

                                                   v_EPS_ContextStatus_Tx,

                                                   v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType,

                                                   p_ForcedAttach);

        /* Step 8: Release RRC connection */

        if (p_ReleaseRequired == rrcConnectionRelease) { // @sic R5s110127, R5s150830 sic@

          f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

        }

      } else { // No bearer established so follow table 2; @sic R5-110746 sic@

        if (ischosen (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST)) {   // @sic R5-123734 sic@

          //   Step 4: The SS transmits a TRACKING AREA UPDATE REJECT message with EMM cause = "No EPS Bearer Activated".

          v_Rel8_9 := true; // @sic R5-123734 sic@

          SRB.send(cas_SRB1_NasPdu_REQ(p_CellId,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                    cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_Cause_NoEPSBearerActive))));

          // Step 5: SS waits for tsc_WaitforAttach seconds to receive the Attach Request on the existing RRC Connection

          t_WaitforAttach.start;

          alt {

          [] SRB.receive(car_SRB1_NasPdu_IND(p_CellId,

                                             cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                                               cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                               cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType))))

            -> value v_ReceivedAsp

            {

              v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

              t_WaitforAttach.stop;

            }

          [] t_WaitforAttach.timeout

            {

              // Step 5 : The SS releases the RRC connection.

              f_EUTRA_RRC_ConnectionRelease(p_CellId);   /* @sic R5-110803 change 13 sic@ */

              // Step 6:

              if (not pc_Auto_PDN_Connectivity) { // @sic R5-120754, R5-125117 sic@

                f_UT_RequestAdditionalPDNConnectivity (UT, 6); // @sic R5-125116 sic@

              }

              v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                         tsc_RRC_TI_Def,

                                                         cr_EstablishmentCause_Any,

                                                         cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),

                                                                                           cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                           cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)));

            }

          }

        } // end of TAU received, so Attach Req must have been received first

        // Activate DHCP and ICMPv6 and reset IMS if previously registered to IMS

        f_EUTRA_ResetIPandIMS (p_CellId); // @sic R5s150913 sic@

        // Steps 7 - 22

        if (p_ReleaseRequired == rrcConnectionRelease) { // @sic R5-160947 sic@

          f_EUTRA_InitialRegistration_Step5_18b1(p_CellId, STATE2_IDLEUPDATE, p_ForcedAttach, v_NasInd);  // @sic R5s110395 sic@

        } else {

          f_EUTRA_InitialRegistration_Step5_16c1(p_CellId, STATE2_IDLEUPDATE, p_ForcedAttach, v_NasInd);  // @sic R5s110395 sic@

        }

        if (not ispresent (v_Rel)) { // @sic R5-123734 sic@

          v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease (); // can get value now, so can now check behaviour correct for release of UE

          if ((not(valueof(v_Rel) == rel8 or valueof(v_Rel) == rel9)) and v_Rel8_9 ) { // @sic R5s120659 sic@ @sic R5s130275 sic@

            f_SetVerdictInconc(__FILE__, __LINE__, "Rel8/9 behaviour not allowed for higher release UE");

          }

        }

      }

      return v_NasCountUL; // @sic R5s120774 sic@

  }  
